
Ball-Indentation of Multi-layered Foils

This Measurement Note introduces a novel finite element (FE)solution to the elastic
indentation of a multi-layered coated half-plane. It describes a method of predicting
elastic indentation behaviour that can be used to check validity of experimental mea-
surements. The problem was analysed using the commercial finite element software
ANSYS, for which a macro was written in order to automatise the calculations, using
the ANSYS Parametric Design Language (APDL). This macro wasused to determine
the effects of the number of layers and their relative thickness, of the friction between
the indenter and the top layer of the coating, and of the residual stress state on the elas-
tic response of the system. The inputs to the FE model were reduced so that the user
needs only concentrate on the thermo-elastic properties ofthe layers and the substrate,
the exact thickness of the layers, and the stress free temperature of the structure. All
inputs normally used a database of referenced thermo-mechanical properties, based
on values determined experimentally, although manual input in the macro remained
an option. It was further shown that friction could be neglected, whereas residual
stresses, introduced via the stress-free temperature, influenced to a great extent the
elastic response of the coated system. An alternative approach to finite elements,
based on recent semi-analytical solutions to the stress state and the penetration of the
indenter, was further suggested for future developments.

1. Introduction

This document describes the contents and the use of a
macro for Finite Element solution of the axisymmetric
nano-indentation of multi-layered systems. The macro
is written in the ANSYS Parametric Design Language
(APDL), and presents a fully automated alternative to
compute the elastic response of a multi-layered system
indented by a diamond sphere. The macro is called
with a number of arguments that enable the definition
of the substrate and coating materials, the number of
bi-layers constituting the coating and their average in-
dividual thickness, as well as the indenter radius. The
geometry of the problem and the mesh were optimised
so that the user is required to provide a minimum of
inputs. A range of materials are available for the sub-
strate and the coating, and their mechanical properties
are extracted from a database linked to the macro.

After a brief review of the state of the art, the geom-
etry and the FE solution are described. Application ex-
amples, emphasizing the effects of the number of layers,
of the friction between indenter and coating surfaces, as
well as residual stresses, are further discussed.

2. Background

2.1. Models of elastic indentation

Nano-indentation has long been promoted as the ideal
tool for the characterisation of materials on the nano-
scale [1, 2]. The ability to load the material over very
small areas, in the range of a few tens of square nanome-
ters, together with its load control capabilities, down to
theµN, is bound to enable the local determination of the
mechanical properties of the material’s surface. A thor-
ough review was written recently [3], and the principles
of nano-indentation will not be repeated here. However,
the analysis of nano-indentation data remains far from
being simple, and it is useful to complete the discussion
of ref. [3] by arguments related to multi-layered geome-
tries. As discussed in ref. [4], a number of unknowns
clearly hamper the determination of elastic constants by
nano-indentation in multi-layered systems. The extent
of the effect of friction between the indenter and the
surface, of residual stresses, and the experimentally un-
resolved contact area, to name only a few, are still issues
of great importance, that pose the question: can elastic
properties of coatings be determined by means of in-
dentation? The measurement of an elastic property is
based on the determination of the contact stiffnessS,
that relates to the area of contactA by:

whereEt, νt andEs, νs are the elastic modulus and
Poisson’s ratio of the tip and sample, respectively. It is
important to note that the validity of Equation1 has not
been established on the basis of experimental results for
nano-indentation, although other nano-mechanical con-
tacts have been determined based on the use of, e.g.,
surface force apparatus. Most of the mathematical mod-
elling of elastic contacts is based on the original works
of Hertz [5] and Boussinesq [6]. Their models consider
the contact of two axisymmetric curved surfaces, taken
as axisymmetric half-spaces, where the contact region
is small compared to the radius of curvature of the con-
tacting surfaces. By letting one radius tend towards in-
finity, the geometry for the traditional indentation of a
plane by a sphere is obtained, as shown in Fig.1. Fol-
lowing the Hertz model, the radius of the contact region
a relates to the radiusR of the elastic sphere indenting
the surface and the applied loadP as:

The displacementδ of the sphere into the surface being
given by:

Although the Hertz model is an exact solution for ho-
mogeneous materials, it has no equivalent for multi-
layered systems, so that a FE approach was preferred.

2.2. Indentation of multi-layers

The geometry of the problem is the indentation by a
hard sphere (the indenter) of an elastic half-plane that
is coated by a large number of elastic layers with thick-
ness in the sub-micrometer range. The diameter of the
indenter typically ranges between 5µm and 100µm, al-
though in practice smaller radii can be used, and usually
the largest diameter for testing nano-structured coatings
is about 25µm. The thickness of the individual coating
layers is between a few nanometers and the microme-
ter. One of the difficulties of such simulation is the large
variation of aspect ratio when the thickness of the layers
reaches a few nanometers, together with the uncertainty
of their mechanical properties at this scale. Invariably,
the large aspect ratio of the coating layers leads to a
large number of elements and, as a consequence, re-
source intensive models. Several meshing strategies can
be used to reduce the number of elements, with relative
success. In this study it was found that a good choice
of dimensions enables the use of a number of elements
consistent with the use of a standard PC, without par-
ticular need for an intricate mesh (although the mesh
dimensions are only chosen to fit the purpose of most
hard coatings on more compliant substrates). As such,
the length of the substrate was chosen to be equal to
twenty times the thickness of the thinnest coating layer
times the ratio of the largest to the smallest modulus;

the thickness of the substrate was one half of this quan-
tity. It was further checked that reaction forces on the
boundaries were independent of changes to larger val-
ues, within computer error (10−12). Simulations were
performed for layers of thickness larger in all cases than
the limit of 5-10 nm, where it is thought that molecu-
lar modelling should be used instead. Here, the thick-
ness of layers was kept above 50 nm, where it has been
demonstrated experimentally that traditional mechanics
are valid [4]. Guidance on model requirements was ob-
tained from earlier analytical and FE works at NPL of
indentation of single-layered systems [7].

3. Presentation of the APDL macro

In order to deal properly with materials at the small
scale, the macro was written for use with the set of con-
sistent units [µm, g, nN, kPa, s,◦C]. The file begins
with an automatic clearing of any previous results, mak-
ing sure that the memory buffer is empty, followed by
an instruction that toggles on the multi-processor capa-
bility. A number of definitions of parameters follow, in-
cluding the load step history, and the number of entries
in the materials database used. The substrate length and
thickness are then defined, based on the smallest thick-
ness of coating layer.

After these initialisations have been performed, the
/PREP7 environment is accessed. The materials data
array is created, and populated with the content of the
file DATAMAT.DAT: contents of this file and references
to the source of the data are found in the associated file
DATAMAT.TXT. The density and elastic modulus are
converted to the set of units in use. Poisson’s ratio and
coefficient of thermal expansion are non-dimensional
quantities and therefore only need to be read. The in-
denter material properties are fixed at the moment, set
to the properties of diamond. The friction coefficient
between the indenter and the top layer of the coating is
left as a parameter which can be tuned (it defaults to 0).
The materials properties are shown in Table1.

The next step involves building the finite element
model. Here, the task is fully automated through the use
of the macro. Upon introduction of the right parameters,
one of the two configurations will be chosen, namely
[substrate/AB...AB] or [substrate/AB...ABA]. The geom-
etry is built by defining keypoints first, joining them

Table 1: Thermo-mechanical properties of materials
used in FE simulations

Material E (GPa) ν α (µm/m/K)
Steel 200 0.29 17.28

Titanium 108 0.37 8.90
Titanium Nitride 600 0.20 9.40

Diamond 1200 0.10 1.18

Figure 1: View of the finite element mesh

with lines, and then assigning an area number to the
newly created layer. The material property is then at-
tributed to the layer, and the layer is meshed. A free
meshing was used, and all the constituents were meshed
by PLANE183 elements, which are 2D axisymmetric
8-nodes quadrilateral solid elements (Figure1). Once
the discrete geometry has been obtained, the boundary
conditions are fixed. In order to obtain an axisymmetric
geometry, they-axis is defined as a cylindrical symme-
try axis, by having no displacements in thex-direction.
The bottom of the substrate is clamped. A tempera-
ture boundary condition is further applied to all nodes
of the model, in order to simulate the effect of a residual
stress state. This temperature effectively corresponds to
a stress free temperature, that is calculated for materials
with constant thermo-elastic properties in the tempera-
ture range considered. While this is not true in general
for any physical system, it has been shown to be a rea-
sonable numericalapproximation, for cases where it is
possible to account for viscoelastic and other time de-
pendent phenomena by adjusting∆T accordingly [8].

Two options of ANSYS are used when solving this
problem:

• gap closure: we did not use the fixed penalty stiff-
ness to determine the area of contact, but fix the
initial gap to zero, using theICONT instruction.
The contact stiffness is further calculated and used
as a starting point for regression;

• non-linear geometry: although displacements are
small, turning on the non-linear geometry option
has been shown to help the convergence of the
problem in similar situations.

Finally, for each load step we record the maximum
indentation depth, and save the results database. The
displacementδ versus loadP tables are further exported
to an ASCII file for easy representation.

3



4. APDL Macro

!===================================================================================================
!== NANO-INDENTATION OF A MULTI-LAYERED COATING ON SUBSTRATE BY A SPHERICAL INDENTER ==
!== ALL UNITS IN kPa, um, g, s, nN WHERE NEEDED ==
!== ARGUMENT 1: SUBSTRATE MATERIAL (INTEGER) ==
!== ARGUMENT 2: COATING MATERIAL A (INTEGER) ==
!== ARGUMENT 3: COATING MATERIAL B (INTEGER) ==
!== ARGUMENT 4: NUMBER OF BI-LAYERS (INTEGER) ==
!== ARGUMENT 5: LAY-UP OF THE COATING (1) OR (2) ==
!== ARGUMENT 6: AVERAGE THICKNESS OF LAYERS A (REAL) ==
!== ARGUMENT 7: AVERAGE THICKNESS OF LAYERS B (REAL) ==
!== ARGUMENT 8: INDENTER RADIUS (REAL) ==
!== ARGUMENT 9: FRICTION COEFFICIENT BETWEEN INDENTER AND TOP FILM (REAL, VALUE BETWEEN [0;1]) ==
!== ARGUMENT 10: THERMAL GRADIENT (SIMULATION OF RESIDUAL STRESS) BETWEEN TOP LAYER AND SUBSTRATE ==
!===================================================================================================
/CLEAR
/CONFIG,NPROC,2
/TITLE, AXISYMMETRIC NANO-INDENTATION
/COM, ANALYSIS PARAMETERS
NMATER=9 ! NUMBER OF MATERIALS IN DATABASE
PI=3.14159265359
RIND=ARG8 ! RADIUS OF SPHERICAL INDENTER

*IF,ARG6,LE,ARG7,THEN
LX=20.*ARG7 ! LENGTH OF SUBSTRATE

*ELSE
LX=20.*ARG6

*ENDIF
LY=0.5*LX ! THICKNESS OF SUBSTRATE
ELEMDIVY1=3 ! DEFAULT NUMBER OF ELEMENTS ACROSS SMALLEST THICKNESS
ARFILM=10 ! ASPECT RATIO OF ELEMENTS TO MESH FILM
TGAP=0. ! GAP CLOSURE FOR ICONT OPTION
NSTEPS=20

*DIM,FORCEVERT,ARRAY,1+NSTEPS ! DEFINE ARRAY TO STORE APPLIED FORCE
*DIM,DISPLVERT,ARRAY,1+NSTEPS ! DEFINE ARRAY TO STORE RESULTING PENETRATION

*DO,I,2,1+NSTEPS/2
FORCEVERT(I) = 10000.*(I-1)

*ENDDO

*DO,I,2+NSTEPS/2,1+NSTEPS
FORCEVERT(I) = 100000.*(I-NSTEPS/2)

*ENDDO
/PREP7
/COM, MATERIALS DATA
*DIM,MATDATA,,NMATER,4

*VREAD,MATDATA(1,1),DATAMAT,DAT,,JIK,4,NMATER
(4F7.1)
MP,DENS,1,MATDATA(ARG1,3)*1E-15 ! MATERIAL 1 DENSITY
MP,EX,1,MATDATA(ARG1,1)*1.E6 ! MATERIAL 1 YOUNG’S MODULUS
MP,PRXY,1,MATDATA(ARG1,2) ! MATERIAL 1 POISSON’S RATIO
MP,ALPX,1,MATDATA(ARG1,4)*1.E-6 ! MATERIAL 1 CTE
!!!.... INSERT OTHER MATERIALS HERE ......
MP,MU,9,ARG9 ! SURFACE CONTACT PROPERTIES WITH INDENTER
TREF,0
/COM, FINITE ELEMENT MODEL
ANTYPE,STATIC ! STATIC ANALYSIS
MSHAPE,0,2D ! MESH USING QUAD ELEMENTS
MSHKEY,1 ! USE FREE MESHING
ET,1,PLANE183,,,1 ! PLANE183 (UX,UY) AXISYMMETRIC 2D QUAD SOLID P-ELEMENT
ET,2,169 ! TARGET SURFACE ELEMENT
ET,3,172,0,,,,1 ! CONTACT SURFACE ELEMENT

*IF,ARG6,LE,ARG7,THEN
ELEMDIVY2=NINT(ARG7*ELEMDIVY1/ARG6)

*ELSE
ELEMDIVY2=ELEMDIVY1
ELEMDIVY1=NINT(ARG6*ELEMDIVY2/ARG7)

*ENDIF
ELEMDIVX1=NINT(LX*ELEMDIVY1/(ARFILM*ARG6))
ELEMDIVX2=ELEMDIVX1
RECTNG,0,LX,0,LY ! DEFINE RECTANGLE OF SUBSTRATE, CYLINDER AFTER ROTATION
LESIZE,1,,,ELEMDIVX1
LESIZE,3,,,ELEMDIVX1
LESIZE,2,,,ELEMDIVX1*2
LESIZE,4,,,ELEMDIVX1*2
TYPE,1 ! SELECT ELEMENT TYPE
AATT,1 ! ASSIGN TYPE TO AREA
AMESH,1 ! MESH AREA
HTOT=LY
KLSTART=3
*DO,INOD,1,ARG4 ! DEFINE FILM STACKING

HTOT=HTOT+ARG6
K,(1+4*INOD),LX,HTOT
K,(2+4*INOD),0,HTOT
HTOT=HTOT+ARG7
K,(3+4*INOD),LX,HTOT
K,(4+4*INOD),0,HTOT
L,KLSTART,(1+4*INOD)
L,(1+4*INOD),(2+4*INOD)

L,(2+4*INOD),(KLSTART+1)
LESIZE,6*INOD-1,,,ELEMDIVY1 ! ASSIGN DIVISIONS THROUGH THICKNESS
LESIZE,6*INOD,,,ELEMDIVX1 ! ASSIGN DIVISIONS ACROSS LENGTH
LESIZE,6*INOD+1,,,ELEMDIVY1 ! ASSIGN DIVISIONS THROUGH THICKNESS
A,KLSTART,(1+4*INOD),(2+4*INOD),(1+KLSTART) ! ASSIGN AREA NUMBER
AATT,2
AMESH,(2*INOD)
KLSTART=1+4*INOD
L,KLSTART,(3+4*INOD)
L,(3+4*INOD),(4+4*INOD)
L,(4+4*INOD),(KLSTART+1)
LESIZE,6*INOD+2,,,ELEMDIVY2 ! ASSIGN DIVISIONS THROUGH THICKNESS
LESIZE,6*INOD+3,,,ELEMDIVX2 ! ASSIGN DIVISIONS ACROSS LENGTH
LESIZE,6*INOD+4,,,ELEMDIVY2 ! ASSIGN DIVISIONS THROUGH THICKNESS
A,KLSTART,(3+4*INOD),(4+4*INOD),(1+KLSTART) ! ASSIGN AREA NUMBER
AATT,3
AMESH,(2*INOD+1)
KLSTART=3+4*INOD

*ENDDO
NSEL,S,LOC,Y,HTOT
NSEL,R,LOC,X,0

*GET,NODEN0,NODE,0,NUM,MAX
HTOT=HTOT+RIND+TGAP ! CALCULATE HTOT FOR TOP OF GEOMETRY
K,4*ARG4+5,0,HTOT ! KEYPOINT CORNER CIRCULAR INDENTER
CIRCLE,4*ARG4+5,RIND,,6,90,1 ! DEFINE QUARTER CIRCLE INDENTER
L,4*ARG4+5,4*ARG4+6 ! CLOSE QUARTER CIRCLE, LEFT
L,4*ARG4+7,4*ARG4+5 ! CLOSE QUARTER CIRCLE, TOP
ELEMDIVCIRC=3*NINT(ELEMDIVX1*PI*RIND/(2*LX))

*IF,MOD(ELEMDIVCIRC,2),EQ,1,THEN
ELEMDIVCIRC=ELEMDIVCIRC+1

*ENDIF
LESIZE,6*ARG4+5,,,ELEMDIVCIRC,2 ! ASSIGN DIVISIONS ON 1/4 CIRCLE
LESIZE,6*ARG4+6,,,ELEMDIVCIRC ! ASSIGN DIVISIONS VERTICAL EDGE
LESIZE,6*ARG4+7,,,ELEMDIVCIRC ! ASSIGN DIVISIONS TOP EDGE
AL,6*ARG4+5,6*ARG4+7,6*ARG4+6 ! ASSIGN AREA 4 THROUGH LINES
AATT,3 ! ASSIGN DIAMOND PROPERTIES TO INDENTER
AMESH, 2+2*ARG4 ! MESH AREA 4
TYPE,2 ! --- CONTACT AREAS ---// TARGET SURFACE
MAT,9 ! FRICTION COEFFICIENT
LSEL,S,,,6*ARG4+3 ! TOP SURFACE, TOP ELECTRODE
NSLL,S,1 ! SELECT ALL NODES
ESURF ! ASSIGN TARGET TO SURFACE
TYPE,3 ! CONTACT SURFACE
LSEL,S,,,6*ARG4+5 ! QUARTER CIRCLE (TOP)
NSLL,S,1 ! SELECT ALL NODES
ESURF ! ASSIGN CONTACT TO SURFACE
ALLSEL, ALL
NLGEOM, ON
SAVE,ICONT,db1
RESUME, ICONT,db1
/COM, BOUNDARY CONDITIONs:
NSEL,S,LOC,X,0 ! DEFINE OY AS SYMMETRY AXIS (OX = 0)
D,ALL,UX,0 ! ZERO HORIZONTAL DISPLACEMENT
NSEL,S,LOC,Y,0 ! SELECT BOTTOM SURFACE
D,ALL,UY,0
D,ALL,UX,0
NSEL, ALL ! SELECT ALL NODES
BFUNIF,TEMP,AR10 ! APPLY TEMPERATURE ON ALL NODES
/SOLUTION ! EXIT PREP7, GET INTO SOLVER
SOLVE ! INITIAL TEMPERATURE STEP SOLVE
/POST1 ! EXIT SOLVER< ENTER POST
*GET,DVERTINI,NODE,NODEN0,U,Y ! RETRIEVE VERTICAL RESIDUAL DISPLACEMENT
DISPLVERT(1)=ABS(DVERTINI)
FORCEVERT(1)=0.
/SOLUTION ! RE-ENTER SOLVER

*DO,ISTEP,2,1+NSTEPS
NSEL,S,LOC,Y,HTOT ! SELECT TOP SURFACE
D,ALL,UX,0 ! HORIZONTAL DISPLACEMENT CONSTRAINED
F,ALL,FY,-FORCEVERT(ISTEP)/ELEMDIVX1 ! APPLY FORCE BOUNDARY CONDITION
NSEL,ALL
SOLVE ! SOLVE STATIC PROBLEM
NSEL,S,LOC,Y,HTOT

*GET,NNDS,NODE,,COUNT
ND=0
UAVE=0
*DO,I,1,NNDS

ND=NDNEXT(ND)
*GET,ULOC,NODE,ND,U,Y
UAVE=UAVE+ULOC

*ENDDO
UAVE=UAVE/NNDS
DISPLVERT(ISTEP)=ABS(UAVE)

*ENDDO

*CFOPEN,LOADDISPL,DAT ! OPEN FILE LOADDISPL.DAT FOR WRITING
*VWRITE,DISPLVERT(1),FORCEVERT(1) ! WRITE RESULTS
(E12.5,’ ’,E12.5)

*CFCLOS ! CLOSE FILE LOADDISPL.DAT

5. Results

5.1. Number of layers and relative thickness

The first set of results that need to be examined is the
influence of the number of layers and their thickness.
Two sets of layer thickness were used: 500 nm/100 nm
and 250 nm/50 nm. Load versus displacement curves
for such [Ti, TiN]N bi-layers on steel, with N equal to
5 or 10, are shown in Figure2. Firstly, all curves follow
the trend given by Equation (4) upon adjustment ofEr,
i.e. they are all approximated within less than 0.01%
error by a power law of exponent 3/2. This is not sur-
prising, as a purely elastic response is calculated, which
corresponds in all boundaries to the Hertzian case. No
easily identifiable quantity, such as the composite mod-
ulus in bending or tension, could be extracted from the
contact stiffness determined by fitting these curves to
Eq. (4), though. It was found that the problem cannot
be inverted easily, and resolving individual layer prop-
erties from a load-displacement curve need additional
numerical regression to be coupled to the FE solution.

Secondly, the four plots displayed show a dominant
influence of the bending stiffness of the hardest layer,
TiN in this case: the closer to the substrate and the
thicker the top layer, the stiffer the contact. This is
shown in Fig. 3 as well, where the large variation in
vertical stressσz occurs in the first Ti layer: the stiffer
TiN layer above it buffers the indenter’s vertical stress
by spreading it away from the contact zone. In Fig.3,
this is illustrated by the zone of maximum stress extend-
ing to about 5 times the area of contact.

Other specific features of this model, not discussed
here, include the stress transfer singularity at the inter-
faces between layers: above three elements in the thick-
ness of the layers, it can be seen on, e.g. a plot of shear
stress against the vertical axis, which shows large dis-
continuities between layers.

Figure 2: Load-displacement curves for various number
of layers of Ti,TiN bi-layers, as indicated on the Figure

Figure 3: Vertical stress distribution in a [Ti,TiN]5 sys-
tem on a steel substrate

This type of problem, involving more than one mil-
lion elements at least, was deemed too resource inten-
sive for this study, where the main objective remains
the characterisation of layers and composite behaviour,
rather than their interfaces.

5.2. Effects of friction

Next, it was decided to examine the effects of friction
between the indenter and the top layer on the stress re-
sponse. The reason is associated with the experimental
uncertainty arising when the indenter is pushed inside
a surface: up to now, it has not been possible to mea-
sure the contact area while the indenter is in contact,
and therefore contact models need to be used in order
to approximate the contact behaviour. Figure4 displays

Figure 4: Relative difference of displacement obtained
for various friction coefficientsµ

the relative difference between displacements obtained
with and without friction, that is(δµ−δ)/δ. The friction
coefficients of 0.04 and 0.1 were chosen because the
former was experimentally determined as static friction
in a bending experiment [4] with similar surfaces, and
the latter is close to values obtained for dynamic fric-
tion from nano-scratch tests. Surprisingly, it was not
possible to distinguish between load versus displace-
ment curves, so that the type of representation of Fig.
4 was preferred instead. The maximum difference, for
µ = 0.04, is never greater than 0.2%, and never ex-
ceeds 0.6% forµ = 0.1. These simulations therefore
indicate that friction between the indenter and the top
layer has a negligible effect on the load-displacement
curve, well within the experimental error of such mea-
surement. Consequently, when the top layer is a smooth
metal or ceramic, friction can be neglected.

Apart from that important fact, it is interesting to
notice that the relative difference increases sharply (in
an absolute value sense) during the early loading stages,
which correspond to the largest increase in contact area,
and then stabilises when the contact area becomes al-
most constant.

5.3. Effects of residual stresses

Last, but not least, the effects of residual stresses were
examined. It is well known, in polymer-based multi-
layered systems, that residual stresses can change not
only the shape of the composite, but also the internal
stress state, to the point that internal damage can be sus-
tained during demoulding of structures. During the past
ten years, conclusive measurements of residual stresses,
and in particular the internal stress level in thin layers,
have been obtained for various material combinations,
and shown to reach up to several GPa for diamond-like
coatings. The point is, even though the internal stress
state is essentially tensile within the layers, it is enough
to have a large influence on the through-thickness com-
ponent of stress.

It is not surprising, therefore, to witness a large shift
in the load-displacement curves when a temperature dif-
ference is applied first, prior to loading of the structure
by the spherical indenter (Fig.5). This effectively gen-
erates a change of the structural response of the coating,
which is in effect pre-loaded. In the present case, be-
cause the coefficient of thermal expansion of the coating
layers is smaller than that of the substrate, when the ma-
terial is cooled-down the coating is put in tension, with
a stiffness smaller than would be obtained by indention
only. The effect of residual stresses is a decrease of the
secant stiffness, particularly in the early contact stages.
Notice as well that the tangent stiffness, changed by al-
most a factor ten at low loads, tends to be similar when
high enough loads, in the order of 0.5 mN, are reached.

To conclude, the effect of internal stresses, which
changes the coated material’s response enormously, can-
not be neglected when analysing indentation results. Sim-

Figure 5: Load-displacement curves with or without
residual stresses, as indicated by the stress free temper-
ature on the Figure

ilarly, it is recommended that residual stresses are mea-
sured independently, e.g. by means of transmission elec-
tron microscopy of ion-milled sections.

6. Future Work

To summarise the arguments above, it was concluded
that large variations in the elastic indentation behaviour
were obtained when the number and relative thickness
of the coating layers were changed, and also when the
residual stress state was accounted for. The effects of
friction between the indenter and the top layer, within
a reasonable range restricted to hard coatings and dia-
mond indenter, were found to be negligible.

As a result, we claim that the behaviour of the com-
posite under indentation can be determined by carefully
specifying the dimensions of the problem, the thermo-
elastic properties, and the stress free temperature. Stem-
ming from these assumptions, finding a closed-form so-
lution to the problem is easier, and although no such
working has been proposed comprehensively, the fol-
lowing approach is relevant. Starting from the works of
Shield and Bogy, Green’s tensor describing continuity
of stress and displacement between layers is determined
[9]. This approach makes use of Hankel transforms and
is followed by a cascade elimination, so that all applied
boundary conditions (top layer and substrate) are linked
in the same equation. Shield and Bogy’s approach was
for a flat, cylindrical indenter. A similar argument has
been used to solve the spherical indentation problem by
Aizikovich et al. [10], who obtain an analytical solution
for a graded half space, that can be extended to a coated
half-space. This approach, leading to an integral equa-
tion, could not be solved, thus being of limited use for
solving experimental contact problems.

Very recently, however, Perriot and Barthel have re-formulated the analysis to enable the problem to be solved
at low numerical cost [11]. It is recommended that this latter approach is followed, after being completed by
including the residual stress term.

7. Conclusions

In this work, finite element solutions for the elastic indentation of a multi-layered coated half-plane have been
devised and used to determine the effects of the number of layers, of the friction between the indenter and the top
layer of the coating, and of the residual stress state. The inputs to the FE model were reduced so that the user
needs only concentrate on the thermo-elastic properties ofthe layers and the substrate, the exact thickness of the
layers, and the stress free temperature. It was further shown that friction could be neglected, whereas residual
stresses influence to a great extent the elastic response of the coated system. An alternative approach, based on
recent semi-analytical solutions to the stress state and the penetration of the indenter, was suggested.

